



**Micro
Technology
Unlimited**

K-1002-3 SOURCE

P.E.T. MUSIC SOFTWARE
FOR P.E.T. 2001

MAY 1979

1979
COPYRIGHT NOTICE

This, user's manual, and all program listings in this package are copyrighted. The user or customer may make BACKUP copies only to protect against loss. The copyright notices must be added to and remain intact on all such backup copies.

The programs may be used only on the computer systems owned directly by the customer himself and may not be reproduced and shipped with systems sold or rented by the customer.

Micro Technology Unlimited
PO Box 4596
Manchester, NH 03108

PETMS PET - DAC MUSIC SYSTEM
DOCUMENTATION AND EQUATES

```

3      ;      THIS PROGRAM IS AN ADAPTATION TO THE PET OF THE MICRO
4      ;      TECHNOLOGY UNLIMITED 4 PART HARMONY MUSIC PROGRAM FOR THE
5      ;      KIM-1 AND 8-BIT DAC. IT CONSISTS OF A MAIN SEQUENCER PROGRAM
6      ;      AND SEVERAL SUBROUTINES. THE SEQUENCER SCANS THROUGH A
7      ;      SEQUENCE TABLE STARTING AT SEQTAB AND INTERPRETS INSTRUCTIONS
8      ;      TO PLAY SONG TABLES, CREATE WAVEFORMS, CHANGE TEMPO, AND
9      ;      CHANGE THE NUMBER OF VOICES PLAYING. THE SEQUENCER THEN CALLS
10     ;      THE APPROPRIATE ROUTINE TO CARRY OUT THE SPECIFIED OPERATION.
11
12     ;      THIS PROGRAM IS INTENDED TO BE CALLED FROM BASIC USING THE SYS
13     ;      COMMAND. IT WILL RETURN TO BASIC WHEN THE END OF THE SEQUENCE
14     ;      TABLE (DENOTED BY FFFF) IS FOUND.
15
16     ;      THIS PROGRAM CONTAINS A PAGE ZERO SWAP ROUTINE TO ALLOW USE OF
17     ;      PAGE ZERO WITHOUT WIPING OUT PET BASIC.
18
19
20 E84F      USRPA      =      X'E84F      ; USER PORT A DATA REGISTER
21 E843      USRPAD     =      X'E843      ; USER PORT A DIRECTION REGISTER
22 0F00      SEQTAB     =      X'0F00      ; ADDRESS OF BEGINNING OF SEQUENCE TABLE
23
24 0000      .,=      0      ; KEEP ALL CONSTANTS AND DATA ON PAGE ZERO
25
26 0000      V1LL:      .,+=      1      ; FRACTIONAL PART OF VOICE 1 WAVEFORM
27 0001 000B      V1L:      .WORD      X'0B00      ; TABLE ADDRESS; BYTE AND PAGE PART
28 0003      V2LL:      .,+=      1      ; FRACTIONAL PART OF VOICE 2 WAVEFORM
29 0004 000C      V2L:      .WORD      X'0C00      ; TABLE ADDRESS; BYTE AND PAGE PART
30 0006      V3LL:      .,+=      1      ; FRACTIONAL PART OF VOICE 3 WAVEFORM
31 0007 000D      V3L:      .WORD      X'0D00      ; TABLE ADDRESS; BYTE AND PAGE PART
32 0009      V4LL:      .,+=      1      ; FRACTIONAL PART OF VOICE 4 WAVEFORM
33 000A 000E      V4L:      .WORD      X'0E00      ; TABLE ADDRESS; BYTE AND PAGE PART
34 000C      V1INCL:     .,+=      1      ; FRACTIONAL PART OF VOICE 1 WAVEFORM
35 000D      V1INCH:     .,+=      1      ; TABLE ADDRESS INCREMENT; INTEGER PART
36 000E      V2INCL:     .,+=      1      ; FRACTIONAL PART OF VOICE 1 WAVEFORM
37 000F      V2INCH:     .,+=      1      ; TABLE ADDRESS INCREMENT; INTEGER PART
38 0010      V3INCL:     .,+=      1      ; FRACTIONAL PART OF VOICE 1 WAVEFORM
39 0011      V3INCH:     .,+=      1      ; TABLE ADDRESS INCREMENT; INTEGER PART
40 0012      V4INCL:     .,+=      1      ; FRACTIONAL PART OF VOICE 1 WAVEFORM
41 0013      V4INCH:     .,+=      1      ; TABLE ADDRESS INCREMENT; INTEGER PART
42
43 0014      DUR:      .,+=      1      ; DURATION PARAMETER
44 0015      SNGAL:      .,+=      1      ; WORKING SONG ADDRESS LOW
45 0016      SNGAH:      .,+=      1      ; WORKING SONG ADDRESS HIGH
46 0017      SNGALO:      .,+=      1      ; SONG ADDRESS LOW
47 0018      SNGAHO:      .,+=      1      ; SONG ADDRESS HIGH
48 0019      VOICEN:     .,+=      2      ; WORKING SEQUENCE TABLE ADDRESS
49 001B 000C      INCA:      .WORD      X'0C00      ; INITIAL OFFSET FOR NOTE STORAGE
50 001D 50      TEMPO:      .BYTE      80      ; TEMPO PARAMETER
51 001E 14      VLAST:      .BYTE      X'14      ; STORES PARAMETER THAT CONTROLS NUMBER OF
52      ;      VOICES PLAYING
53

```

PETMS PET - DAC MUSIC SYSTE
NOTE FREQUENCY TABLE

```

54      .PAGE 'NOTE FREQUENCY TABLE'
55      ; NOTE FREQUENCY TABLE FOR 8.77 KHZ SAMPLE RATE. RANGE FROM C1 TO C7

56 001F 0000 NOTAB: .BYTE X'00,X'00 ; 00 REST
57 0021 01E9 .BYTE X'01,X'E9 ; 02 C2
58 0023 0206 .BYTE X'02,X'06 ; 04 C2#
59 0025 0225 .BYTE X'02,X'25 ; 06 D2
60 0027 0245 .BYTE X'02,X'45 ; 08 D2#
61 0029 0268 .BYTE X'02,X'68 ; 0A E2
62 002B 028C .BYTE X'02,X'8C ; 0C F2
63 002D 02B3 .BYTE X'02,X'B3 ; 0E F2#
64 002F 02DC .BYTE X'02,X'DC ; 10 G2
65 0031 0308 .BYTE X'03,X'08 ; 12 G2#
66 0033 0336 .BYTE X'03,X'36 ; 14 A2
67 0035 0367 .BYTE X'03,X'67 ; 16 A2#
68 0037 039A .BYTE X'03,X'9A ; 18 B2
69 0039 03D1 .BYTE X'03,X'D1 ; 1A C3
70 003B 040B .BYTE X'04,X'0B ; 1C C3#
71 003D 0449 .BYTE X'04,X'49 ; 1E D3
72 003F 048A .BYTE X'04,X'8A ; 20 D3#
73 0041 04CF .BYTE X'04,X'CF ; 22 E3
74 0043 0519 .BYTE X'05,X'19 ; 24 F3
75 0045 0566 .BYTE X'05,X'66 ; 26 F3#
76 0047 05B8 .BYTE X'05,X'B8 ; 28 G3
77 0049 0A0F .BYTE X'0A,X'0F ; 2A G3#
78 004B 066C .BYTE X'06,X'6C ; 2C A3
79 004D 06CD .BYTE X'06,X'CD ; 2E A3#
80 004F 0735 .BYTE X'07,X'35 ; 30 B3
81 0051 07A3 .BYTE X'07,X'A3 ; 32 C4
82 0053 0817 .BYTE X'08,X'17 ; 34 C4#
83 0055 0892 .BYTE X'08,X'92 ; 36 D4
84 0057 0915 .BYTE X'09,X'15 ; 38 D4#
85 0059 099F .BYTE X'09,X'9F ; 3A E4
86 005B 0A31 .BYTE X'0A,X'31 ; 3C F4
87 005D 0ACC .BYTE X'0A,X'CC ; 3E F4#
88 005F 0B71 .BYTE X'0B,X'71 ; 40 G4
89 0061 0C1F .BYTE X'0C,X'1F ; 42 G4#
90 0063 0CD7 .BYTE X'0C,X'D7 ; 44 A4
91 0065 0D9B .BYTE X'0D,X'9B ; 46 A4#
92 0067 0E6A .BYTE X'0E,X'6A ; 48 B4
93 0069 0F45 .BYTE X'0F,X'45 ; 4A C5
94 006B 102E .BYTE X'10,X'2E ; 4C C5#
95 006D 1124 .BYTE X'11,X'24 ; 4E D5
96 006F 1229 .BYTE X'12,X'29 ; 50 D5#
97 0071 133E .BYTE X'13,X'3E ; 52 E5
98 0073 1462 .BYTE X'14,X'62 ; 54 F5
99 0075 1599 .BYTE X'15,X'99 ; 56 F5#
100 0077 16E2 .BYTE X'16,X'E2 ; 58 G5
101 0079 183E .BYTE X'18,X'3E ; 5A G5#
102 007B 19AF .BYTE X'19,X'AF ; 5C A5
103 007D 1B36 .BYTE X'1B,X'36 ; 5E A5#
104 007F 1CD4 .BYTE X'1C,X'D4 ; 60 B5
105 0081 1E8B .BYTE X'1E,X'8B ; 62 C6
106 0083 205C .BYTE X'20,X'5C ; 64 C6#
107 0085 2248 .BYTE X'22,X'48 ; 66 D6

```

PETMS PET - DAC MUSIC SYSTE
NOTE FREQUENCY TABLE

108	0087	2452	.BYTE	X'24,X'52	; 68	D6#
109	0089	267B	.BYTE	X'26,X'7B	; 6A	E6
110	008B	28C5	.BYTE	X'28,X'C5	; 6C	F6
111	008D	2B31	.BYTE	X'2B,X'31	; 6E	F6#
112	008F	2DC3	.BYTE	X'2D,X'C3	; 70	G6
113	0091	307C	.BYTE	X'30,X'7C	; 72	G6#
114	0093	335E	.BYTE	X'33,X'5E	; 74	A6
115	0095	366C	.BYTE	X'36,X'6C	; 76	A6#
116	0097	39A8	.BYTE	X'39,X'A8	; 78	B6
117	0099	3D15	.BYTE	X'3D,X'15	; 7A	C7

118

119 ; LOW OCTAVE EXTENSION OF FREQUENCY TABLE

120						
121	009B	00F4	.BYTE	X'00,X'F4	; 7C	C1
122	009D	0103	.BYTE	X'01,X'03	; 7E	C1#
123	009F	0112	.BYTE	X'01,X'12	; 80	D1
124	00A1	0123	.BYTE	X'01,X'23	; 82	D1#
125	00A3	0134	.BYTE	X'01,X'34	; 84	E1
126	00A5	0146	.BYTE	X'01,X'46	; 86	F1
127	00A7	015A	.BYTE	X'01,X'5A	; 88	F1#
128	00A9	016D	.BYTE	X'01,X'6D	; 8A	G1
129	00AB	0184	.BYTE	X'01,X'84	; 8C	G1#
130	00AD	019B	.BYTE	X'01,X'9B	; 8E	A1
131	00AF	01B3	.BYTE	X'01,X'B3	; 90	A1#
132	00B1	01CD	.BYTE	X'01,X'CD	; 92	B1

133

134 ; ADDITIONAL DATA STORAGE

135						
136	00B3	00	SEQL:	.BYTE	0	; LOW BYTE OF POINTER TO SEQUENCE TABLE
137	00B4	0F	SEQH:	.BYTE	X'0F	; HIGH BYTE OF POINTER TO SEQUENCE TABLE
138	00B5		M2:	.=.+	1	; MULTIPLICAND USED IN MULTFR
139	00B6		M1:	.=.+	1	; MULTIPLIER USED IN MULTFR
140	00B7			.=.+	1	; BYTE NOT USED
141	00B8		AMPL:	.=.+	1	; HARMONIC AMPLITUDE DURING FOURIER CALC
142	00B9		HARM:	.=.+	1	; HARMONIC NUMBER TO BE PROCESSED
143	00BA		WAVTBL:	.=.+	1	; WAVEFORM TABLE POINTER, LOW BYTE
144	00BB		WAVTBH:	.=.+	1	; WAVEFORM TABLE POINTER, HIGH BYTE
145	00BC		SINEL:	.=.+	1	; LOW PART OF SINE TABLE LOOKUP ADDRESS
146	00BD	09	SINEH:	.BYTE	X'09	; HIGH PART OF SINE TABLE LOOKUP ADDRESS
147	00BE		PHASE:	.=.+	1	; PHASE ANGLE OF HARMONIC DURING FOURIER
148	00BF	3F	MAXAMP:	.BYTE	X'3F	; MAXIMUM ALLOWABLE AMPLITUDE FOR COMPUTED
149						; WAVEFORM
150	00C0		FCOEFF:	.=.+	38	; STORAGE FOR FOURIER COEFFICIENTS
151						

PETMS PET - DAC MUSIC SYSTE
INITIALIZATION AND MAIN PROGRAM

```

152      ;      .PAGE 'INITIALIZATION AND MAIN PROGRAM'
153      ;      THIS IS THE ENTRY POINT FOR THE MUSIC SYSTEM.
154      ;      CONTROL IS PASSED HERE FROM BASIC BY KEYING IN RUN AFTER
155      ;      LOADING FROM TAPE OR BY A SYS COMMAND (SYS 1899).
156      ;      CONTROL IS RETURNED TO BASIC BY COMPLETING A SONG SEQUENCE OR
157      ;      HITTING THE SHIFT KEY
158 00E6      . =      X'76B      ; START IMMEDIATELY AFTER THE MACHINE
159      ; LANGUAGE MONITOR
160
161 076B 78      MAIN:  SEI      ; PREVENT KEYBOARD SCAN INTERRUPTS
162 076C 20DE07      JSR      SWCHR ; SWITCH DATA - PAGE ZERO
163 076F A9FF      LDA      #'FF ; ...SET USER PORT WITH DAC CONNECTED TO
164 0771 8D43E8      STA      USRPAD ; ALL OUTPUT...
165 0774 D8      CLD      ; ALL ARITHMETIC IN BINARY
166 0775 20270A      JSR      SEQNCR ; SEQUENCER TAKES OVER
167
168 0778 20DE07      RENTRY: JSR      SWCHR ; SWITCH DATA BACK - PAGE ZERO
169 077B A2FC      LDX      #'FC ; ...SET STACK POINTER FOR
170 077D 9A      TXS      ; CLEAN REENTRY INTO BASIC...
171 077E 60      RTS      ; BACK TO BASIC
172

```

PETMS PET - DAC MUSIC SYSTE
SONG TABLE INTERPRETER

```

.PAGE 'SONG TABLE INTERPRETER'
173      ; SONG TABLE INTERPRETER
174      ; THIS ROUTINE IS CALLED BY THE SEQUENCER TO SCAN THROUGH A
175      ; SONG TABLE AND PLAY THE INDICATED NOTES.
176      ; ENCOUNTERING A ZERO DURATION SPECIFICATION FORCES A RETURN
177
178 077F A517 MUSIC: LDA SNGALO ; ...FETCH INITIAL ENTRY
179 0781 8515 STA SNGAL ; POINT OF SONG DATA AND
180 0783 A518 LDA SNGAHO ; STORE IN SONGAL, SONGAH...
181 0785 8516 STA SNGAH
182
183 0787 58 FETCH: CLI ; ALLOW KEYBOARD SCAN
184 0788 AD0402 LDA X'0204 ; ...IF SHIFT KEY DEPRESSED,
185 078B D0EB BNE RENTRY ; BRANCH BACK TO REENTRY...
186 078D 78 SEI ; PREVENT INTERRUPTS
187 078E A000 LDY #0 ; GET SET FOR INDIRECT ADDRESSING
188 0790 A51B LDA INCA
189 0792 8519 STA VOICEN
190 0794 B115 LDA (SNGAL),Y ; FIRST BYTE IN SONG DATA IS DURATION
191 0796 F03C BEQ DONE ; DONE IF DUR = 0
192 0798 C901 CMP #1 ; ...X'01 SIGNIFIES A JUMP IN
193 079A F029 BEQ JUMP ; THE SONG DATA...
194 079C 8514 STA DUR ; STORE IT IN DUR
195 079E E615 INC SNGAL ; ...GET SET TO PICK
196 07A0 D002 BNE LOAD ; UP A NOTE FROM
197 07A2 E616 INC SNGAH ; SONG TABLE...
198
199 07A4 B115 LOAD: LDA (SNGAL),Y ; LOAD A NOTE
200 07A6 AA TAX ; ...USE AS INDEX TO
201 07A7 B520 LDA NOTAB+1,X ; PICK UP TWO BYTES
202 07A9 9119 STA (VOICEN),Y ; FROM NOTE TABLE...
203 07AB E619 INC VOICEN ; ...GET SET TO
204 07AD B51F LDA NOTAB,X ; PICK UP NEXT
205 07AF 9119 STA (VOICEN),Y ; NOTE FROM SONG
206 07B1 E615 INC SNGAL ; TABLE...
207 07B3 D002 BNE INCVOC
208 07B5 E616 INC SNGAH
209 07B7 E619 INCVOC: INC VOICEN ; NEXT NOTE
210 07B9 A519 LDA VOICEN ; ...IF NOT ALL
211 07BB C51E CMP VLAST ; VOICES ARE USED UP
212 07BD D0E5 BNE LOAD ; BRANCH TO LOAD...
213 07BF 20FC08 JSR PLAY ; THEN PLAY THE PHRASE
214 07C2 4C8707 JMP FETCH ; AND BACK TO FETCH
215
216 07C5 C8 JUMP: INY ; ...FETCH NEXT
217 07C6 B115 LDA (SNGAL),Y ; TWO BYTES FROM
218 07C8 48 PHA ; SONG DATA AND
219 07C9 C8 INY ; RESET SONGAL, SONGAH
220 07CA B115 LDA (SNGAL),Y ; FOR THE JUMP...
221 07CC 8516 STA SNGAH
222 07CE 68 PLA
223 07CF 8515 STA SNGAL
224 07D1 4C8707 JMP FETCH ; BACK TO GET NEXT DURATION
225 07D4 60 DONE: RTS ; RETURN WHEN 0 DURATION SEEN
226

```

PETMS PET - DAC MUSIC SYSTE
ZEROER, SWTCHR, MULTFR

```

227          ; .PAGE 'ZEROER, SWTCHR, MULTFR'
228          ; ZEROER - SUBROUTINE TO ZERO 256 BYTES STARTING AT THE CURRENT
229          ;          CONTENTS OF WAVTBL,WAVTBH
230 07D5 A900  ZEROER: LDA    #0          ; ZERO A
231 07D7 A8    TAY          ; ZERO Y
232 07D8 91BA  AGAIN1: STA    (WAVTBL),Y ; ...ZERO ALL ITEMS IN
233 07DA 88    DEY          ; WAVTBL,WAVTBH...
234 07DB D0FB  BNE    AGAIN1 ; REPEAT 256 TIMES
235 07DD 60    RTS          ; DONE
236
237
238          ; SWTCHR - SUBROUTINE TO SWAP PAGE ZERO UP THROUGH X'F6
239          ; WITH THE CONTENTS OF DATA
240
241 07DE A2F7  SWTCHR: LDX    #'F7
242 07E0 B5FF  AGAIN2: LDA    X'FF,X      ; FETCH DATA FROM PAGE ZERO
243 07E2 48    PHA          ; CACHE ON STACK
244 07E3 BDEF07 LDA    DATA-1,X ; ...FETCH OTHER DATA AND
245 07E6 95FF  STA    X'FF,X      ; TRANSFER TO PAGE ZERO...
246 07E8 68    PLA          ; ...FETCH OLD DATA FROM
247 07E9 9DEF07 STA    DATA-1,X ; STACK AND SAVE IT...
248 07EC CA    DEX
249 07ED D0F1  BNE    AGAIN2 ; REPEAT
250 07EF 60    RTS          ; DONE
251
252 07F0      DATA:  .=.+  X'F7      ; STORAGE FOR IMAGE OF PAGE ZERO
253
254
255          ; MULTFR - 8 BY 8 UNSIGNED FRACTIONAL MULTIPLICATION SUBROUTINE
256          ; MULTIPLIES CONTENTS OF M1 BY CONTENTS OF M2 AND RETURNS WITH
257          ; PRODUCT IN ACCUMULATOR
258
259
260 08E7 A900  MULTFR: LDA    #0          ; ZERO THE PRODUCT
261 08E9 06B6  SHIFT: ASL    M1          ; ...SHIFT M1 TO PUT MSB IN
262 08EB 9006  BCC    ZERO          ; CARRY FLAG, BRANCH IF CARRY CLEAR...
263 08ED 46B5  LSR    M2          ; ...SHIFT M2 AND
264 08EF 65B5  ADC    M2          ; ADD TO A...
265 08F1 D0F6  BNE    SHIFT        ; THEN BRANCH BACK
266 08F3 F006  ZERO:  BEQ    DONE1     ; DONE IF RESULT IS ZERO
267 08F5 46B5  LSR    M2          ; ...SHIFT M2 TO GET LSB IN
268 08F7 B0F0  BCS    SHIFT        ; CARRY FLAG, BRANCH IF CARRY SET
269 08F9 D0EE  BNE    SHIFT        ; OR IF RESULT IS NOT EQUAL TO ZERO...
270 08FB 60    DONE1: RTS          ; DONE
271

```

PETMS PET - DAC MUSIC SYSTE
PLAY - SOUND GENERATION ROUTINE

```

272          ;          .PAGE 'PLAY - SOUND GENERATION ROUTINE'
273          ;          SOUND GENERATION ROUTINE
274          ;          LIFTED INTACT FROM AN ARTICLE BY HAL CHAMBERLIN IN THE 9/77
275          ;          ISSUE OF BYTE MAGAZINE
276          ;          SOUNDS 4 SIMULTANEOUS WAVEFORMS AT A SAMPLE RATE OF 8.77KHZ
277 08FC A000      PLAY:  LDY    #0          ; GET SET FOR STRAIGHT INDIRECT
278 08FE A61D      LDX    TEMPO          ; GET TEMPO IN X
279 0900 18        CNTNUE: CLC          ; ...ADD UP INTEGER PARTS
280 0901 B101      LDA    (V1L),Y        ; OF WAVEFORM POINTERS
281 0903 7104      ADC    (V2L),Y        ; AND DUMP THE RESULT
282 0905 7107      ADC    (V3L),Y        ; TO THE USER PORT...
283 0907 710A      ADC    (V4L),Y
284 0909 8D4FE8    STA    USRPA
285 090C A500      LDA    V1LL          ; ...THEN GET SET FOR NEXT
286 090E 650C      ADC    V1INCL         ; WAVEFORM POINTS FOR ALL
287 0910 8500      STA    V1LL          ; FOUR VOICES...
288 0912 A501      LDA    V1L
289 0914 650D      ADC    V1INCH
290 0916 8501      STA    V1L
291 0918 A503      LDA    V2LL
292 091A 650E      ADC    V2INCL
293 091C 8503      STA    V2LL
294 091E A504      LDA    V2L
295 0920 650F      ADC    V2INCH
296 0922 8504      STA    V2L
297 0924 A506      LDA    V3LL
298 0926 6510      ADC    V3INCL
299 0928 8506      STA    V3LL
300 092A A507      LDA    V3L
301 092C 6511      ADC    V3INCH
302 092E 8507      STA    V3L
303 0930 A509      LDA    V4LL
304 0932 6512      ADC    V4INCL
305 0934 8509      STA    V4LL
306 0936 A50A      LDA    V4L
307 0938 6513      ADC    V4INCH
308 093A 850A      STA    V4L
309 093C CA        DEX
310 093D D008      BNE    TMWAS          ; ...WHEN TEMPO COUNTER TIMES
311 093F C614      DEC    DUR            ; OUT, DECREMENT DURATION...
312 0941 F00C      BEQ    DNPLAY         ; DONE IF DURATION TIMED OUT
313 0943 A61D      LDX    TEMPO          ; ...RESTORE TEMPO AND
314 0945 D0B9      BNE    CNTNUE         ; BRANCH BACK...
315
316 0947 D000      TMWAS:  BNE    TMWAS2      ; ...WASTE 12 STATES BEFORE
317 0949 D000      TMWAS2: BNE    TMWAS3      ; BRANCHING BACK (TO PRESERVE
318 094B D000      TMWAS3: BNE    TMWAS4      ; SYMETRY)...
319 094D D0B1      TMWAS4: BNE    CNTNUE
320 094F 60        DNPLAY: RTS          ; DONE
321

```

PETMS PET - DAC MUSIC SYSTE
SINE TABLE

```

322          ; .PAGE 'SINE TABLE'
323          ; FIRST QUADRANT SINE TABLE
324          ; VALUES FOR OTHER QUADRANTS ARE OBTAINED BY SYMMETRY
325          ; ENTRIES ARE POSITIVE BINARY FRACTIONS
326 0950 80838689 SINE: .BYTE X'80,X'83,X'86,X'89
327 0954 8C909396 .BYTE X'8C,X'90,X'93,X'96
328 0958 999C9FA2 .BYTE X'99,X'9C,X'9F,X'A2
329 095C A5A8ABAE .BYTE X'A5,X'A8,X'AB,X'AE
330 0960 B1B3B6B9 .BYTE X'B1,X'B3,X'B6,X'B9
331 0964 BCBFC1C4 .BYTE X'BC,X'BF,X'C1,X'C4
332 0968 C7C9CCCE .BYTE X'C7,X'C9,X'CC,X'CE
333 096C D1D3D5D8 .BYTE X'D1,X'D3,X'D5,X'D8
334 0970 DADCDEEO .BYTE X'DA,X'DC,X'DE,X'EO
335 0974 E2E4E6E8 .BYTE X'E2,X'E4,X'E6,X'E8
336 0978 EAEBEDEF .BYTE X'EA,X'EB,X'ED,X'EF
337 097C F0F1F3F4 .BYTE X'F0,X'F1,X'F3,X'F4
338 0980 F5F6F8F9 .BYTE X'F5,X'F6,X'F8,X'F9
339 0984 FAFAFBFC .BYTE X'FA,X'FA,X'FB,X'FC
340 0988 FDFDFEFE .BYTE X'FD,X'FD,X'FE,X'FE
341 098C FFFFFFFF .BYTE X'FE,X'FF,X'FF,X'FF
342 0990 FF .BYTE X'FF
343

```

PETMS PET - DAC MUSIC SYSTE
FOURIER SERIES WAVEFORM COMPUTATION ROUTINE

```

344      ; .PAGE 'FOURIER SERIES WAVEFORM COMPUTATION ROUTINE'
345      ; FOURIER SERIES WAVEFORM COMPUTATION ROUTINE
346      ; SCANS THROUGH A LIST OF FOURIER COEFFICIENTS STARTING AT
347      ; FCOEFF AND CREATES A WAVEFORM AT THE PAGE ADDRESS SPECIFIED
348      ; BY THE CONTENTS OF WAVTABL+1.
349      ; EACH COEFFICIENT CONSISTS OF 3 BYTES, FIRST IS HARMONIC NUMBER
350      ; SECOND IS AMPLITUDE, AND THIRD IS PHASE ANGLE. A HARMONIC
351      ; NUMBER OF ZERO SIGNIFIES END OF LIST
352 0991 A2C0 FORIER: LDX #FCOEFF ; ADDRESS OF BEGINNING OF COEFFICIENT TABL
353 0993 B500 NXHARM: LDA O,X ; ...FIRST FETCH A HARMONIC NUMBER
354 0995 F043 BEQ MAXMIN ; (A ZERO SIGNIFIES END OF LIST) AND
355 0997 85B9 STA HARM ; STORE IT...
356 0999 E8 INX ; ...NEXT BYTE IS
357 099A B500 LDA O,X ; AMPLITUDE...
358 099C 85B8 STA AMPL
359 099E E8 INX ; ...NEXT BYTE IS
360 099F B500 LDA O,X ; PHASE ANGLE...
361 09A1 85BE STA PHASE
362 09A3 84BA STY WAVTABL ; INITIALIZE WAVTABL TO 0
363
364 09A5 A5BE REPEAT: LDA PHASE ; FETCH PHASE ANGLE
365 09A7 257F AND X'7F ; MASK OFF BIT 7
366 09A9 C940 CMP #X'40 ; CHECK FOR GTX'40 (90 DEG)
367 09AB 3004 BMI STORE ; ...IF GT X'40, MASK
368 09AD 293F AND #X'3F ; OFF BIT 6 AND COMPLEMENT
369 09AF 493F EOR #X'3F ; CARRY SET (TO CORRECT SECOND AND
370 ; THIRD QUADRANT) AND ADD
371 ; LOW SINE TABLE ADDRESS TO GET CORRECT
372 09B1 6950 STORE: ADC #SINE&X'FF ; REGION OF SINE TABLE...
373 09B3 85BC STA SINEL ; ...NOW FETCH THE SINE
374 09B5 B1BC LDA (SINEL),Y ; OF THE ADJUSTED ANGLE...
375 09B7 24BE BIT PHASE ; CHECK BIT 7 OF ORIGINAL ANGLE
376 09B9 1005 BPL STORE2 ; ...IF ORIGINAL ANGLE WAS GT 180 DEG,
377 09BB 49FF EOR #X'FF ; TAKE COMPLEMENT OF SINE...
378 09BD 18 CLC
379 09BE 6901 ADC #1
380 09C0 85B6 STORE2: STA M1 ; ...STORE SINE (PHASE ANGLE) AND
381 09C2 A5B8 LDA AMPL ; AMPLITUDE TO PREPARE FOR THE
382 09C4 85B5 STA M2 ; MULTIPLICATION...
383 09C6 20E708 JSR MULTFR ; DO THE MULTIPLICATION
384 ; ...THEN BUILD UP THE
385 09C9 71BA ADC (WAVTABL),Y ; WAVEFORM...(CARRY CLEARED)
386 09CB 91BA STA (WAVTABL),Y
387 09CD A5B9 LDA HARM ; FETCH HARMONIC NUMBER
388 ; ...ADD HARMONIC NUMBER
389 09CF 65BE ADC PHASE ; TO PHASE ANGLE TO GET SET
390 09D1 85BE STA PHASE ; FOR NEXT ENTRY...
391 09D3 E68A INC WAVTABL
392 09D5 D0CE BNE REPEAT ; REPEAT PROCESS 256DEC TIMES
393 09D7 E8 INX ; ...THEN GET SET TO FETCH
394 09D8 D0B9 BNE NXHARM ; NEXT SET OF FOURIER COEFFICIENTS...
395
396 ; NOW SCAN THROUGH THE COMPUTED WAVEFORM AND FIND THE MAXIMUM
397 ; AND MINIMUM SAMPLE VALUES

```

PETMS PET - DAC MUSIC SYSTE
FOURIER SERIES WAVEFORM COMPUTATION ROUTINE

```

398
399 09DA AA      MAXMIN: TAX      ; ...SET MAXIMUM=0 IN M1
400 09DB 85B6    STA      M1      ; AND MINIMUM=FF IN M2...
401 09DD CA      DEX
402 09DE 86B5    STX      M2
403
404 09E0 B1BA    REPT2: LDA      (WAVTBL),Y ; FETCH A BYTE FROM WAVTAB
405 09E2 C5B5    CMP      M2      ; ...IF LESS THAN PREVIOUS MINIMUM,
406 09E4 8002    BCS      NEXT1    ; PUT NEW MINIMUM IN M2...
407 09E6 85B5    STA      M2
408 09E8 C5B6    NEXT1: CMP      M1      ; ...IF GREATER THAN PREVIOUS MAXIMUM,
409 09EA 9002    BCC      NEXT2    ; PUT NEW MAXIMUM IN M1...
410 09EC 85B6    STA      M1
411 09EE C8      NEXT2: INY      ; ...THEN REPEAT PROCESS ON
412 09EF D0EF    BNE      REPT2    ; ENTIRE WAVEFORM TABLE...
413
414      ;      NOW SHIFT THE WAVEFORM DOWN SO THAT THE MINIMUM SAMPLE IS ZERO
415
416 09F1 B1BA    REPT3: LDA      (WAVTBL),Y ; ...SUBTRACT MINIMUM FROM
417 09F3 38      SEC      ; EVERY ENTRY IN THE WAVEFORM
418 09F4 E5B5    SBC      M2      ; TABLE...(CARRY SET)
419 09F6 91BA    STA      (WAVTBL),Y
420 09F8 C8      INY
421 09F9 D0F6    BNE      REPT3
422
423      ;      DIVIDE (MAX-MIN) BY MAXAMP TO GET SCALE FACTOR
424
425 09FB A5B6      LDA      M1      ; ...SUBTRACT MINIMUM FROM
426      ; MAXIMUM TO GET THE NEW
427 09FD E5B5      SBC      M2      ; MAXIMUM...
428 09FF 85B6      STA      M1
429 0A01 A208      LDX      #8      ; ...GET SET TO CALCULATE
430      ; SCALE FACTOR = MAXAMP/MAXIMUM...
431 0A03 A5BF      LDA      MAXAMP  ; USE MAXIMUM ALLOWABLE AMPLITUDE AS
432      ; THE NUMERATOR
433      ; AND M1 AS THE DENOMINATOR
434      ; AND DO A DIVISON
435 0A05 0A      SHIFT2: ASLA      ; PUT MSB OF A INTO CARRY FLAG
436 0A06 B004      BCS      SUBTR    ; ...BRANCH TO ROLL IF CARRY IS CLEAR
437 0A08 C5B6      CMP      M1      ; AND IF A IS LESS THAN DENOMINATOR...
438 0A0A 9003      BCC      ROLL
439 0A0C E5B6      SUBTR: SBC      M1      ; SUBTRACT DENOMINATOR FROM A
440 0A0E 38      SEC      ; PUT A 1 IN CARRY FLAG
441 0A0F 26B8      ROLL:  ROL      AMPL  ; SHIFT THE CARRY BIT INTO QUOTIENT
442 0A11 CA      DEX      ; DO IT EIGHT TIMES
443 0A12 D0F1      BNE      SHIFT2
444
445      ;      FINALLY SCALE ALL SAMPLES SO THAT THE MAXIMUM = MAXAMP
446
447 0A14 84BA      STY      WAVTBL  ; INITIALIZE WAVETABLE (Y=0)
448 0A16 A5B8      REPT4: LDA      AMPL  ; ...FETCH SCALE FACTOR AND
449 0A18 85B5      STA      M2      ; WAVEFORM POINT, THEN MULTIPLY...
450 0A1A B1BA      LDA      (WAVTBL),Y
451 0A1C 85B6      STA      M1
452 0A1E 20E708    JSR      MULTFR

```

PETMS PET - DAC MUSIC SYSTE
FOURIER SERIES WAVEFORM COMPUTATION ROUTINE

453	0A21	91BA	STA	(WAVTBL),Y	; STORE THE CORRECTED POINT
454	0A23	88	DEY		; ...SCALE ALL THE POINTS IN
455	0A24	D0F0	BNE	REPT4	; THE WAVEFORM TABLE
456	0A26	60	RTS		; DONE WITH FOURIER SYNTHESIS
457					

PETMS PET - DAC MUSIC SYSTE
SEQUENCER - SCAN THROUGH SEQUENCE TABLE

```

.PAGE 'SEQUENCER - SCAN THROUGH SEQUENCE TABLE'
458 ; SEQUENCER - SCANS THROUGH THE SEQUENCE TABLE AND INTERPRETS
459 ; THE COMMANDS SEEN.
460
461 0A27 A900 SEQNCR: LDA #0 ; ...INITIALIZE Y, SET START
462 0A29 85B3 STA SEQL ; ADDRESS OF SEQUENCER TABLE
463 0A2B A8 TAY ; TO BE X'0F00...
464 0A2C A90F LDA #SEQT/256
465 0A2E 85B4 STA SEQH
466
467 0A30 B1B3 NEXCMD: LDA (SEQL),Y ; FETCH A BYTE FROM SEQUENCE TABLE
468 0A32 C9FF CMP #X'FF ; ...IF IT IS =X'FF, THEN A SPECIAL
469 0A34 F012 BEQ ACTION ; ACTION IS INDICATED...
470 0A36 8518 STA SNGAHO ; OTHERWISE IT IS HIGH ADDRESS OF SONG
471 0A38 C8 INY ; ...NEXT BYTE IS LOW ADDRESS
472 0A39 B1B3 LDA (SEQL),Y ; OF SONG...
473 0A3B 8517 STA SNGALO
474 0A3D 98 TYA ; SAVE Y ON STACK
475 0A3E 48 PHA
476 0A3F 207F07 JSR MUSIC ; PLAY THE SONG SEGMENT
477 0A42 68 PLA ; ...FETCH Y
478 0A43 A8 TAY ; FROM STACK...
479 0A44 C8 NEXEN: INY ; ...THEN BRANCH TO FETCH
480 0A45 D0E9 BNE NEXCMD ; THE NEXT COMMAND...
481 0A47 60 RTS ; RETURN IF SEQUENCE TABLE EXCEEDS ONE
482 ; PAGE OF MEMORY
483
484 0A48 C8 ACTION: INY ; ...CHECK NEXT BYTE IN
485 0A49 B1B3 LDA (SEQL),Y ; SEQUENCING TABLE...
486 0A4B D007 BNE ONE ; BRANCH IF NOT EQUAL TO 0
487 0A4D C8 INY ; ...ZERO MEANS VOICE CHANGE, SO
488 0A4E B1B3 LDA (SEQL),Y ; FETCH IT AND STORE IN APPROPRIATE
489 0A50 851E STA VLAST ; PLACE...
490 0A52 D0F0 BNE NEXEN ; BRANCH BACK
491
492 0A54 C901 ONE: CMP #1 ; IS IT =1?
493 0A56 D010 BNE TWO ; IF NO, BRANCH TO CHECK FURTHER
494 0A58 A202 LDX #2 ; A 1 MEANS CHANGE VOICE ASSIGNMENTS
495 0A5A C8 NEXVCE: INY
496 0A5B B1B3 LDA (SEQL),Y ; ...FETCH THE DATA FROM
497 0A5D 9500 STA O,X ; THE SEQUENCING TABLE AND
498 0A5F E8 INX ; USE FOR VOICE ASSIGNMENTS...
499 0A60 E8 INX
500 0A61 E8 INX
501 0A62 E00E CPX #X'0E
502 0A64 D0F4 BNE NEXVCE
503 0A66 F0DC BEQ NEXEN ; THEN BRANCH BACK
504
505 0A68 C902 TWO: CMP #2 ; IS IT =2?
506 0A6A D007 BNE THREE ; IF NO, BRANCH TO CHECK FURTHER
507 0A6C C8 INY ; ...A 2 MEANS NEXT BYTE IN
508 0A6D B1B3 LDA (SEQL),Y ; SEQUENCING TABLE IS TEMPO...
509 0A6F 851D STA TEMPO ; STORE IT IN TEMPO
510 0A71 D0D1 NEXEN2: BNE NEXEN ; BRANCH BACK FOR NEXT ENTRY
511

```

PETMS PET - DAC MUSIC SYSTE
SEQUENCER - SCAN THROUGH SEQUENCE TABLE

512 0A73 C903	THREE:	CMP	#3	; IS IT =3?
513 0A75 D029		BNE	FOUR	; IF NOT EQUAL TO 3, CHECK FURTHER
514 0A77 C8		INY		; ...A 3 MEANS SYNTHESIZE A
515 0A78 B1B3		LDA	(SEQL),Y	; WAVEFORM, NEXT BYTE IS THE
516 0A7A 85BB		STA	WAVTBH	; PAGE NUMBER...
517 0A7C A2C0		LDX	#FCOEFF&X'FF	; SET UP TO TRANSFER A SET OF COEFFICIENTS
518 0A7E C8	NEXSET:	INY		; ...NEXT BYTE IS HARMONIC
519 0A7F B1B3		LDA	(SEQL),Y	; NUMBER...
520 0A81 9500		STA	0,X	
521 0A83 F00F		BEQ	SYN	; ZERO HARMONIC NUMBER SIGNIFIES END
522 0A85 C8		INY		; ...NEXT TWO BYTES IN
523 0A86 B1B3		LDA	(SEQL),Y	; SEQUENCING TABLE ARE
524 0A88 E8		INX		; AMPLITUDE AND PHASE SHIFT,
525 0A89 9500		STA	0,X	; RESPECTIVELY...
526 0A8B C8		INY		
527 0A8C B1B3		LDA	(SEQL),Y	
528 0A8E E8		INX		
529 0A8F 9500		STA	0,X	
530 0A91 E8		INX		
531 0A92 D0EA		BNE	NEXSET	; FETCH NEXT SET OF COEFFICIENTS
532 0A94 98	SYN:	TYA		; ...SAVE Y ON THE
533 0A95 48		PHA		; STACK...
534 0A96 20D507		JSR	ZEROER	; PUT 0'S IN VOICE PAGE
535 0A99 209109		JSR	FORIER	; COMPUTE THE WAVEFORM AND STORE IT
536 0A9C 68		PLA		; ...FETCH Y BACK
537 0A9D A8		TAY		; FROM STACK...
538 0A9E D0D1	NEXEN3:	BNE	NEXEN2	; THEN BRANCH BACK FOR NEXT ENTRY
539				
540 0AA0 C904	FOUR:	CMP	#4	; IS IT =4?
541 0AA2 D01C		BNE	FIVE	; IF NO, CHECK FURTHER
542 0AA4 C8		INY		; ...A 4 MEANS CREATE A RECTANGULAR
543 0AA5 B1B3		LDA	(SEQL),Y	; WAVEFORM, NEXT BYTE IS PAGE NUMBER...
544 0AA7 85BB		STA	WAVTBH	
545 0AA9 C8		INY		; ...SAVE NEXT Y VALUE
546 0AAA 98		TYA		; ON STACK...
547 0AAB 48		PHA		
548 0AAC B1B3		LDA	(SEQL),Y	; NEXT BYTE IS DUTY CYCLE
549 0AAE 85BE		STA	PHASE	; USE LOCATION PHASE TO STORE IT
550 0AB0 20D507		JSR	ZEROER	; ZERO THE ENTIRE PAGE
551 0AB3 A5BF		LDA	MAXAMP	; FETCH AMPLITUDE
552 0AB5 91BA	NEXT3:	STA	(WAVTBL),Y	; STORE IN WAVEFORM TABLE
553 0AB7 C8		INY		; ...UNTIL Y = DUTY CYCLE
554 0AB8 C4BE		CPY	PHASE	; STORED IN PHASE...
555 0ABA D0F9		BNE	NEXT3	
556 0ABC 68		PLA		; ...GET Y BACK FROM
557 0ABD A8		TAY		; STACK...
558 0ABE D0DE		BNE	NEXEN3	; THEN BRANCH BACK FOR NEXT ENTRY
559				
560 0AC0 C905	FIVE:	CMP	#5	; IS IT =5?
561 0AC2 D00E		BNE	SIX	; IF NOT, CHECK FURTHER
562 0AC4 C8		INY		; ...NEXT BYTE IS WAVEFORM
563 0AC5 B1B3		LDA	(SEQL),Y	; PAGE NUMBER...
564 0AC7 85BB		STA	WAVTBH	; STORE IT
565 0AC9 98		TYA		; ...SAVE Y ON
566 0ACA 48		PHA		; STACK...

PETMS PET - DAC MUSIC SYSTE
SEQUENCER - SCAN THROUGH SEQUENCE TABLE

```

567 OACB 20D507      JSR    ZEROER      ; ZERO THE PAGE
568 OACE 68          PLA                ; ...FETCH Y BACK
569 OACF A8          TAY                ; FROM STACK...
570 OADO DOCC      NEXEN4: BNE    NEXEN3  ; THEN BRANCH BACK FOR NEXT ENTRY
571
572 OAD2 C906      SIX:   CMP    #6      ; IS IT =6?
573 OAD4 D007      BNE    SEVEN         ; IF NOT, CHECK FURTHER
574 OAD6 C8        INY                ; ...NEXT BYTE IS
575 OAD7 B1B3      LDA    (SEQL),Y     ; MAXIMUM AMPLITUDE...
576 OAD9 85BF      STA    MAXAMP
577 OADB DOF3      BNE    NEXEN4      ; THEN BRANCH TO GET NEXT ENTRY
578
579 OADD C907      SEVEN:  CMP    #7      ; IS IT =7?
580 OADF D013      BNE    DONE4         ; IF NOT, SEQUENCING IS DONE
581 OAE1 A90A      LDA    #X'0A        ; ...A 7 MEANS A SUBROUTINE
582 OAE3 48        PHA                ; CALL, FIRST PUSH RETURN ADDRESS FOR
583 OAE4 A943      LDA    #X'43        ; NEXT ENTRY (X'0A44) ONTO STACK...
584 OAE6 48        PHA
585 OAE7 C8        INY                ; ...NEXT TWO BYTES ARE
586 OAE8 B1B3      LDA    (SEQL),Y     ; ADDRESS H AND ADDRESS L
587 OAEA 85C1      STA    FCOEFF+1     ; FOR SUBROUTINE CALL; STORE
588 OAEC C8        INY                ; IN FIRST TWO LOCATIONS OF FOURIER COEFF
589 OAED B1B3      LDA    (SEQL),Y     ; TABLE AND DO AN INDIRECT JUMP...
590 OAEF 85C0      STA    FCOEFF
591 OAF1 6CC000     JMP    (FCOEFF)
592
593 OAF4 60      DONE4:  RTS
594
595
596 0000      .END
NO ERROR LINES

```